

SOLUCIÓN CASO ANÁLISIS FINANCIERO

Deseamos realizar un análisis comparativo de las cuentas consolidadas de dos empresas, concretamente TELEFÓNICA e INDITEX para determinar si responden a nuestras preferencias de inversión. Para ello disponemos de los últimos Balances y Cuentas de Resultados publicados por estas dos empresas en sus Informes Anuales -que en estos momentos son los correspondientes al ejercicio económico 2011-. Queremos elaborar un modelo que calcule determinados ratios y nos muestre un primer diagnóstico de la salud financiera de estas empresas. Asimismo, estudiaremos estos resultados para determinar si nos pudiera interesar invertir en ellas.

Se pide:

1.- Copiar en una hoja de cálculo los Balances de Situación de ambas empresas tanto para el año en curso como los datos comparativos del año anterior. Tan solo tomaremos los datos agregados por masas patrimoniales, de manera que la hoja, antes de introducir los datos numéricos podría quedar como sigue:

	A	B	C	D	E
1	Balance (en miles de euros)	TELEFÓNICA	TELEFÓNICA	INDITEX	INDITEX
2		2011	2010	2011	2010
3	ACTIVO				
4	ACTIVO NO CORRIENTE				
5	ACTIVO CORRIENTE				
6	TOTAL ACTIVO				
7					
8	PATRIMONIO NETO Y PASIVO				
9	PATRIMONIO NETO				
10	PASIVO NO CORRIENTE				
11	PASIVO CORRIENTE				
12	TOTAL PN Y PASIVO				
13					

2.- Una vez introducidos los datos, el primer control que deseamos que realice nuestro modelo consistirá en un control de errores, es decir, asegurarnos de que no se han realizado errores en la introducción de datos, y que por tanto, los Balances están cuadrados, es decir, el Total Activo en cada año es igual al Total Pasivo.

3.- A continuación vamos a realizar un sencillo análisis a través del cálculo de algunos ratios. Concretamente deseamos conocer el estado de liquidez de ambas empresas, su endeudamiento, así como la rentabilidad financiera obtenida en los dos años de análisis.

$$\text{Ratio de liquidez} = \frac{\text{Activo corriente}}{\text{Pasivo corriente}}$$

$$\text{Ratio de Endeudamiento} = \frac{\text{Pasivo}}{\text{PN y Pasivo}}$$

$$\text{Rentabilidad Financiera} = \frac{\text{Rdo. Ejercicio}}{\text{Patrimonio Neto}}$$

4.- En este apartado vamos a calcular el Capital corriente (Fondo de maniobra) de ambas empresas para los dos años estudiados. A continuación deseamos que el modelo indique, mediante una función condicional, la situación de liquidez de la empresa, es decir, si la empresa tiene un nivel de liquidez adecuado o bien si existe un desequilibrio que puede llevar a la empresa a un concurso de acreedores.

$$\text{Capital corriente} = \text{Activo corriente} - \text{Pasivo corriente}$$

5.- Aplicar formatos a la Hoja de análisis mostrando los ratios de endeudamiento y rentabilidad financiera en porcentajes, los valores del balance y la cuenta de resultados en formato euro, etc...

6.- Vamos a hacer un análisis de los ratios de las dos empresas para decidir, en función de una serie de criterios, si el perfil de la empresa responde a nuestras preferencias de inversión. Los criterios son los siguientes:

- El ratio de endeudamiento debe situarse entre el 30% y el 70% -dado que por debajo del 30% se considera demasiado conservador y por encima del 70% se considera excesivamente arriesgado-.
- La rentabilidad financiera debe superar el 10%, al menos en uno de los años de estudio.

En función de estos criterios, junto con la información sobre la situación de liquidez obtenida en el punto 4, resolveremos el apartado en tres pasos:

1º) analizamos si el nivel de endeudamiento es correcto para cada uno de los años en las dos empresas, si está entre los niveles indicados debe aparecer un mensaje de texto que indique “Correcto”, en caso contrario el mensaje deberá indicar “Fallo”.

2º) si la rentabilidad financiera es mayor del 10%, al menos en uno de los dos años, calcularemos la rentabilidad financiera media (promedio de las rentabilidades de los dos años), en caso contrario queremos que muestre el valor 0. Aparecerá un solo valor para cada empresa.

3º) Por último, decidiremos si cada una de estas empresas es apta para realizar o no nuestra inversión en función de que cumpla simultáneamente tres condiciones: que la situación de liquidez sea adecuada en los dos años analizados, que el nivel de endeudamiento en ambos años sea “Correcto” y que la rentabilidad media de los dos años de estudio sea positiva. Aparecerá por tanto una sola celda para cada empresa indicando el mensaje “Apta” si se cumplen las anteriores condiciones y “Problemas” en caso de que no se cumpla alguna de esas condiciones.

Para comprobar que el modelo se ha construido correctamente plantearemos un cambio en los datos que haga modificar los resultados. Supongamos que en los datos de INDITEX en el año 2011 disminuimos la cifra de Patrimonio Neto en 5.000.000 (miles de €) y aumentamos el Pasivo corriente en esa misma cantidad. Si hemos introducido todas las fórmulas correctamente, el modelo se recalculará y obtendremos unos resultados diferentes.

Antes de pasar al apartado siguiente deshacemos los cambios para dejar los datos iniciales.

7.- Dado que estas empresas poseen accionistas extranjeros, consideramos adecuado que el modelo muestre –en una hoja de cálculo distinta- los Balances de ambas empresas en dólares. Con el fin de que el modelo funcione ante cualquier modificación en los datos, **VINCULAREMOS** los balances en dólares a los que ya hemos introducido en euros.

SOLUCIÓN CASO PRÁCTICO ANÁLISIS FINANCIERO

1.- Comenzamos introduciendo el formato de Balance abreviado por masas patrimoniales tal como muestra el enunciado. Podemos obtener la información financiera en las páginas web corporativas de las empresas.

Así, una vez introducidos los datos la hoja quedará así:

	A	B	C	D	E
1	Balance (en miles de euros)	TELEFÓNICA	TELEFÓNICA	INDITEX	INDITEX
2		2011	2010	2011	2010
3	ACTIVO				
4	ACTIVO NO CORRIENTE	108.800.000	108.721.000	5.521.889	4.623.568
5	ACTIVO CORRIENTE	20.823.000	21.054.000	5.437.289	5.202.512
6	TOTAL ACTIVO				
7					
8	PATRIMONIO NETO Y PASIVO				
9	PATRIMONIO NETO	27.383.000	31.684.000	7.455.577	6.423.167
10	PASIVO NO CORRIENTE	69.662.000	64.599.000	800.827	728.005
11	PASIVO CORRIENTE	32.578.000	33.492.000	2.702.774	2.674.907
12	TOTAL PN Y PASIVO				


los valores correspondientes a los Totales Activo y Pasivo no son datos numéricos sino fórmulas, concretamente la suma de los valores correspondientes. Así, en la celda B6, correspondiente al Total Activo de TELEFÓNICA para el año 2011 la fórmula será:

$$=B4+B5$$

ó en su defecto la función.

$$=SUMA(B4:B5)$$

No será necesario introducir la fórmula o función en cada una de las celdas sino que podremos copiarla hacia la derecha. Para ello utilizaremos la opción AUTOLLENADO. Seleccionamos la celda que contiene la fórmula o función que queremos copiar y pulsamos con el ratón el cuadrado negro que aparece en la esquina inferior derecha de la celda activa:




acercamos el ratón hasta que el puntero se convierte en una cruz de trazos gruesos, hacemos clic y, sin soltar, arrastramos hasta la celda E6, al soltar, la fórmula se habrá copiado en todas las celdas del rango. Realizamos los mismos pasos para calcular el Total Pasivo.

Al final, la hoja quedará así:

	A	B	C	D	E
1	Balance (en miles de euros)	TELEFÓNICA	TELEFÓNICA	INDITEX	INDITEX
2		2011	2010	2011	2010
3	ACTIVO				
4	ACTIVO NO CORRIENTE	108.800.000	108.721.000	5.521.889	4.623.568
5	ACTIVO CORRIENTE	20.823.000	21.054.000	5.437.289	5.202.512
6	TOTAL ACTIVO	129.623.000	129.775.000	10.959.178	9.826.080
7					
8	PATRIMONIO NETO Y PASIVO				
9	PATRIMONIO NETO	27.383.000	31.684.000	7.455.577	6.423.167
10	PASIVO NO CORRIENTE	69.662.000	64.599.000	800.827	728.005
11	PASIVO CORRIENTE	32.578.000	33.492.000	2.702.774	2.674.907
12	TOTAL PN Y PASIVO	129.623.000	129.775.000	10.959.178	9.826.079
13					

Antes de pasar al siguiente apartado cambiaremos el nombre a la Hoja de cálculo para asignarle un nombre indicativo de su contenido, la podemos llamar por ejemplo “Balances”. Para ello hacemos doble clic en la pestaña de la Hoja, o bien seleccionamos en el menú FORMATO la opción HOJA/CAMBIAR NOMBRE:



En cualquier caso, la pestaña de la hoja queda seleccionada (en video inverso) y podemos teclear el nombre que deseemos darle a la Hoja, una vez introducido el mismo pulsamos intro para finalizar.

2.- El segundo apartado consiste en comprobar que la introducción de datos ha sido correcta. Para ello vamos a utilizar una **función condicional** muy sencilla que analice la igualdad de los valores contenidos en las celdas que contienen el Total Activo y Total Pasivo.

La función condicional

La función condicional **SI** evalúa una condición establecida en el primer argumento. En caso de que el resultado de la evaluación sea VERDADERO, la celda se iguala al segundo de los argumentos, y en caso de que sea FALSO, al tercero. La función tiene la forma:

`=SI(Condición;Valor si verdadero;Valor si falso)`

La condición debe ser siempre una desigualdad matemática, es decir, podemos comparar si dos celdas son iguales, si una es mayor, mayor o igual, menor, menor o igual que otra, comparar una celda con un texto, etc.

Cuando alguno de los argumentos contiene texto, éste debe ir entrecomillado para que la función lo reconozca como tal.

Si se cumple la condición, es decir si el balance está cuadrado, podemos indicar que aparezca el valor total, y en caso contrario que ponga un cero.

La función en la celda B15 quedaría:

	A	B	
14			
15	introducción de datos correcta ?	=SI(B6=B12;B12;0)	

Una vez aceptada la entrada de la función llenaremos hacia la derecha y quedará:

14					
15	introducción de datos correcta ?	129.623.000	129.775.000	10.959.178	0
16					

Como vemos, Excel ha detectado un error en los datos de INDITEX en el año 2010. Revisamos los datos y comprobamos que, en este caso, no es un error de introducción de datos sino que el error proviene de las propias Cuentas de INDITEX. Al pasar los datos a miles de euros se ha “colado” un error. Nosotros podemos cambiar uno de los datos para que cuadre el Balance, o dejarlo de este modo porque el descuadre es muy pequeño.

3.- Para el cálculo de los ratios de liquidez, endeudamiento y rentabilidad financiera plantearemos las fórmulas correspondientes.

En el caso de la rentabilidad financiera necesitaremos incorporar el dato correspondiente al Resultado del Ejercicio, dato que encontraremos en la Cuenta de Resultados de las respectivas empresas.

Las fórmulas para la empresa TELEFÓNICA en el año 2011 serán:

	A	B
16		
17	Ratio de liquidez	=B5/B11
18		
19	Ratio de endeudamiento	=(B10+B11)/B12
20		
21	Rdo. Ejercicio (dato de la Cta. de resultados)	6187000
22	Rentabilidad financiera	=B21/B9
23		

Una vez copiadas las fórmulas al resto de celdas tendríamos:

	A	B	C	D	E
16					
17	Ratio de liquidez	0,639173675	0,628627732	2,01174386	1,944931917
18					
19	Ratio de endeudamiento	0,78874891	0,755854363	0,3196956	0,346314334
20					
21	Rdo. Ejercicio (dato de la Cta. de resultados)	6.187.000	10.072.000	1.945.532	1.741.280
22	Rentabilidad financiera	0,225943103	0,317889155	0,26094989	0,271093683
23					

4.- Para el cálculo del Capital corriente (Fondo de Maniobra) utilizaremos la misma operativa que en el caso anterior.

Concretamente, en la celda B24 tendremos la fórmula:

$$=B5-B11$$

Además deberemos incluir una condicional que determine si el nivel de liquidez de la empresa es adecuado. **Para ello partiremos de la base de que un valor negativo del Capital corriente Implica posibles dificultades para atender los pagos a corto plazo y por tanto un cierto riesgo de llegar a un concurso de acreedores.**

En este caso la condicional deberá avisarnos de este posible riesgo mediante un mensaje de texto, con lo cual los argumentos 2º y 3º serán textos y por tanto deberán ir entrecomillados.

Por tanto, en B26 introduciremos la siguiente función:

=SI(B24>0;"Adecuada";"Desequilibrio")

5.- Una vez copiada la fórmula del cálculo del Capital corriente y la función condicional a las celdas de la derecha, procederemos a dar formatos a las celdas de forma que el ratio de liquidez aparezca solo con dos decimales y los ratios de endeudamiento y rentabilidad financiera aparezcan en % y también con dos decimales, etc. El modelo completo podría quedar así:

	A	B	C	D	E
1	Balance (en miles de euros)	TELEFÓNICA	TELEFÓNICA	INDITEX	INDITEX
2		2011	2010	2011	2010
3	ACTIVO				
4	ACTIVO NO CORRIENTE	108.800.000,00 €	108.721.000,00 €	5.521.889,00 €	4.623.568,00 €
5	ACTIVO CORRIENTE	20.823.000,00 €	21.054.000,00 €	5.437.289,00 €	5.202.512,00 €
6	TOTAL ACTIVO	129.623.000,00 €	129.775.000,00 €	10.959.178,00 €	9.826.080,00 €
7					
8	PATRIMONIO NETO Y PASIVO				
9	PATRIMONIO NETO	27.383.000,00 €	31.684.000,00 €	7.455.577,00 €	6.423.167,00 €
10	PASIVO NO CORRIENTE	69.662.000,00 €	64.599.000,00 €	800.827,00 €	728.005,00 €
11	PASIVO CORRIENTE	32.578.000,00 €	33.492.000,00 €	2.702.774,00 €	2.674.907,00 €
12	TOTAL PN Y PASIVO	129.623.000,00 €	129.775.000,00 €	10.959.178,00 €	9.826.079,00 €
13					
14					
15	introducción de datos correcta ?	129.623.000,00 €	129.775.000,00 €	10.959.178,00 €	- €
16					
17	Ratio de liquidez	0,64	0,63	2,01	1,94
18					
19	Ratio de endeudamiento	78,87%	75,59%	31,97%	34,63%
20					
21	Rdo. Ejercicio (dato de la Cta. de resultados)	6.187.000,00 €	10.072.000,00 €	1.945.532,00 €	1.741.280,00 €
22	Rentabilidad financiera	22,59%	31,79%	26,09%	27,11%
23					
24	Capital corriente	- 11.755.000,00 €	- 12.438.000,00 €	2.734.515,00 €	2.527.605,00 €
25					
26	Situación de liquidez	Desequilibrio	Desequilibrio	Adecuada	Adecuada
27					

6.- La solución de este apartado se llevará a cabo en tres fases. Antes de empezar y dado que vamos a utilizar los datos de las filas 17, 19 y 22, les habremos aplicado un color de fondo para que resulte más fácil localizar los datos.

	A	B	C	D	E
16					
17	Ratio de liquidez	0,64	0,63	2,01	1,94
18					
19	Ratio de endeudamiento	78,87%	75,59%	31,97%	34,63%
20					
21	Rdo. Ejercicio (dato de la Cta. de resultados)	6.187.000,00 €	10.072.000,00 €	1.945.532,00 €	1.741.280,00 €
22	Rentabilidad financiera	22,59%	31,79%	26,09%	27,11%
23					
24	Capital corriente	- 11.755.000,00 €	- 12.438.000,00 €	2.734.515,00 €	2.527.605,00 €
25					
26	Situación de liquidez	Desequilibrio	Desequilibrio	Adecuada	Adecuada
27					
28					
29	Nivel de Endeudamiento				
30	Rentabilidad total por empresa				
31					
32	Situación de la Empresa				

1º) En primer lugar analizamos para cada uno de los años en cada una de las empresas si el nivel de endeudamiento es el correcto. Para ello, y partiendo de las hipótesis del enunciado deberemos introducir en B29 una función condicional. La condición que queremos analizar es que el nivel de endeudamiento de TELEFÓNICA en el año 2011 (celda B19) esté entre el 30% y el 70%, es decir, en términos matemáticos se trataría de la siguiente expresión:

$$0,3 < B19 < 0,7$$

el problema es que dicha expresión no puede utilizarse en Excel como “condición” de una función condicional. Por ello se tratará de analizar simultáneamente dos condiciones, es decir:

$$B19 > 0,3 \quad \text{y} \quad B19 < 0,7$$

Así, para resolver éste y los siguientes apartados deberemos utilizar funciones condicionales pero con operadores especiales.

Operadores especiales de la función condicional

La función condicional SI puede utilizar operadores especiales, por ejemplo los operadores Y, O. Los argumentos de estos operadores pueden ser a su vez nuevas evaluaciones condicionales, celdas, etc.

Tienen la siguiente forma:

=Y(Condición 1; Condición 2; Condición 3... Condición 30)

=O(Condición 1; Condición 2; Condición 3... Condición 30)

El operador Y devuelve VERDADERO cuando todos sus argumentos son VERDADEROS simultáneamente, mientras que el operador O lo hace cuando alguno de sus argumentos es VERDADERO.

Por tanto la función en B29 será:

$$=SI(Y(B19>0,3;B19<0,7);"Correcto";"Fallo")$$

Y mediante AUTOLLENADO se copiará a las 3 celdas a su derecha hasta E29.

2º) En cuanto al segundo paso, en la celda B30 teclearemos lo siguiente:

$$=SI(O(B22>0,1;C22>0,1);SUMA(B22:C22);0)$$

Y la copiaremos –mediante la opción copiar y pegar- en la celda D30, de manera que aparezca un dato por empresa.

Posteriormente, y con el fin de diferenciar los datos de cada empresa podemos enmarcarlos con un borde.

3º) Por último, decidiremos si la situación financiera de TELEFÓNICA la hace apta para la inversión con las condiciones impuestas por nuestra empresa, para ello en la celda B32 introduciremos la siguiente función condicional:

$$=SI(Y(B26="Adecuada";C26="Adecuada";B29="Correcto";C29="Correcto";B30>0);
"Apta";"Problemas")$$

El resultado será el siguiente:

	A	B	C	D	E
16					
17	Ratio de liquidez	0,64	0,63	2,01	1,94
18					
19	Ratio de endeudamiento	78,87%	75,59%	31,97%	34,63%
20					
21	Rdo. Ejercicio (dato de la Cta. de resultados)	6.187.000,00 €	10.072.000,00 €	1.945.532,00 €	1.741.280,00 €
22	Rentabilidad financiera	22,59%	31,79%	26,09%	27,11%
23					
24	Capital corriente	- 11.755.000,00 €	- 12.438.000,00 €	2.734.515,00 €	2.527.605,00 €
25					
26	Situación de liquidez	Desequilibrio	Desequilibrio	Adecuada	Adecuada
27					
28					
29	Nivel de Endeudamiento	Fallo	Fallo	Correcto	Correcto
30	Rentabilidad total por empresa	27,19%		26,60%	
31					
32	Situación de la Empresa	Problemas		Apta	
33					

Realizamos los cambios propuestos en los datos de INDITEX en el año 2011: disminuimos la cifra de Patrimonio Neto en 5.000.000 (miles de €) y aumentamos el Pasivo Corriente en esa misma cantidad. La hoja de cálculo nos daría los siguientes resultados:

	A	B	C	D	E
16					
17	Ratio de liquidez	0,64	0,63	0,71	1,94
18					
19	Ratio de endeudamiento	78,87%	75,59%	77,59%	34,63%
20					
21	Rdo. Ejercicio (dato de la Cta. de resultados)	6.187.000,00 €	10.072.000,00 €	1.945.532,00 €	1.741.280,00 €
22	Rentabilidad financiera	22,59%	31,79%	79,23%	27,11%
23					
24	Capital corriente	- 11.755.000,00 €	- 12.438.000,00 €	- 2.265.485,00 €	2.527.605,00 €
25					
26	Situación de liquidez	Desequilibrio	Desequilibrio	Desequilibrio	Adecuada
27					
28					
29	Nivel de Endeudamiento	Fallo	Fallo	Fallo	Correcto
30	Rentabilidad total por empresa	27,19%		53,17%	
31					
32	Situación de la Empresa	Problemas		Problemas	
33					

Como vemos, aumenta mucho el endeudamiento y aparecen problemas de liquidez, la empresa pasa a estar calificada como “Problemas”. Es decir, el modelo funciona ante cualquier cambio en los datos, lo que indica que lo hemos construido correctamente.

7.- El último apartado nos pide la conversión de los balances a dólares. Para realizar los cálculos de dicha conversión utilizaremos otra hoja de cálculo distinta, por ejemplo la Hoja2 a la que cambiaremos el nombre, tal como hemos hecho anteriormente con la Hoja1 y la podemos llamar “Balances en dólares”. La solución se realizará **VINCULANDO** los datos desde la hoja de origen.

En primer lugar introducimos en la celda A3 el signo = y hacemos clic en la hoja “Balances” una vez hecho esto vemos que en la barra de escritura aparece lo siguiente:

=Balances!

Al pulsar con el ratón sobre la celda A3 de esta hoja aparece la fórmula:

=Balances!A3

es decir, le estamos indicando a Excel que lo que debe aparecer en la celda A3 de la hoja “Balances en dólares” es lo que contiene la celda A3 de la hoja “Balances”.

A partir de ahí podemos llenar hacia abajo con AUTOLLENADO hasta la celda A11 y las fórmulas quedarán copiadas en el resto de las celdas del modelo.

A continuación haríamos lo mismo para las celdas que contienen el nombre de las empresas y los años para los que estamos trabajando.

Además, en la celda B13 introduciremos el cambio de euros a dólares.

Es muy importante no introducir texto en la celda, solo un dato numérico

Para obtener el tipo de cambio entre divisas podemos consultar alguna página web como por ejemplo:

<http://www.oanda.com>

Suponiendo un tipo de cambio de 1,2973 \$ por euro, el modelo quedaría así:

	A	B	C	D	E
1	Balance (en miles dólares)	TELEFÓNICA	TELEFÓNICA	INDITEX	INDITEX
2		2011	2010	2011	2010
3	ACTIVO				
4	ACTIVO NO CORRIENTE				
5	ACTIVO CORRIENTE				
6	TOTAL ACTIVO				
7					
8	PATRIMONIO NETO				
9	PASIVO NO CORRIENTE				
10	PASIVO CORRIENTE				
11	TOTAL PN Y PASIVO				
12					
13	Tipo de cambio: 1 Euro =	\$1,2973			
14					

Empezaremos calculando el valor en dólares de la primera partida, es decir, el Activo no corriente de TELEFÓNICA en el año 2011. Para ello, tendremos que multiplicar el valor en euros de dicho inmovilizado –dato que aparece en la hoja “Balances”- por el valor del cambio de euros a dólares. Así, en la celda B4 vincularemos la primera parte de la fórmula con la celda B4 de la hoja “Balances” y la multiplicaremos por la celda que contiene el cambio de euros por dólares en la hoja activa. Concretamente la fórmula quedaría:

=Balances!B4*B13

Tras vincular las fórmulas en la celda correspondiente al Activo no corriente habría que hacer lo mismo para cada partida del Balance de cada una de las empresas y en cada uno de los años, sin embargo, Excel nos permite ahorrar tiempo copiando la fórmula en el resto de las celdas hacia abajo y hacia la derecha.

Ahora bien, en este caso debemos tener en cuenta que al copiar fórmulas en Excel éstas operan de forma relativa, es decir, se refieren a las celdas por sus posiciones respecto de la celda que contiene la fórmula, no como una posición absoluta. En nuestro caso, queremos que el primer término de la fórmula vaya cambiando, es decir que cada celda haga referencia a la misma celda de la hoja “Balances”, sin embargo, el segundo término debe contener en todos los casos la misma referencia –en este caso B13- que será la que contenga el cambio de euros a dólares. Por ello, antes de copiar habrá que poner referencias absolutas a esta segunda parte de la fórmula, es decir, conseguir que identifique dicha celda por su posición concreta dentro de la hoja.

Una vez seleccionada la segunda parte de la fórmula y pulsado F4 la fórmula quedará:

=Balances!B4*\$B\$13

Ahora ya podemos copiar a la derecha y hacia abajo mediante la opción autollenado con lo que tendremos calculado el valor en euros de todas las partidas. Las fórmulas serán las siguientes:

	A	B	C	D	E
1	Balance (en miles dólares)	=Balances!B1	=Balances!C1	=Balances!D1	=Balances!E1
2		=Balances!B2	=Balances!C2	=Balances!D2	=Balances!E2
3	=Balances!A3				
4	=Balances!A4	=Balances!B4*\$B\$13	=Balances!C4*\$B\$13	=Balances!D4*\$B\$13	=Balances!E4*\$B\$13
5	=Balances!A5	=Balances!B5*\$B\$13	=Balances!C5*\$B\$13	=Balances!D5*\$B\$13	=Balances!E5*\$B\$13
6	=Balances!A6	=Balances!B6*\$B\$13	=Balances!C6*\$B\$13	=Balances!D6*\$B\$13	=Balances!E6*\$B\$13
7					
8	=Balances!A9	=Balances!B9*\$B\$13	=Balances!C9*\$B\$13	=Balances!D9*\$B\$13	=Balances!E9*\$B\$13
9	=Balances!A10	=Balances!B10*\$B\$13	=Balances!C10*\$B\$13	=Balances!D10*\$B\$13	=Balances!E10*\$B\$13
10	=Balances!A11	=Balances!B11*\$B\$13	=Balances!C11*\$B\$13	=Balances!D11*\$B\$13	=Balances!E11*\$B\$13
11	=Balances!A12	=Balances!B12*\$B\$13	=Balances!C12*\$B\$13	=Balances!D12*\$B\$13	=Balances!E12*\$B\$13
12					
13	Tipo de cambio: 1 Euro =	1,3414			
14					

Los resultados:

	A	B	C	D	E
1	Balance (en miles dólares)	TELEFÓNICA	TELEFÓNICA	INDITEX	INDITEX
2		2011	2010	2011	2010
3	ACTIVO				
4	ACTIVO NO CORRIENTE	\$ 141.142.976,00	\$ 141.040.491,67	\$ 7.163.380,94	\$ 5.998.016,06
5	ACTIVO CORRIENTE	\$ 27.013.053,21	\$ 27.312.722,58	\$ 7.053.631,90	\$ 6.749.062,74
6	TOTAL ACTIVO	\$ 168.156.029,21	\$ 168.353.214,25	\$ 14.217.012,84	\$ 12.747.078,80
7					
8	PATRIMONIO NETO	\$ 35.523.144,41	\$ 41.102.702,68	\$ 9.671.896,37	\$ 8.332.581,85
9	PASIVO NO CORRIENTE	\$ 90.370.422,74	\$ 83.802.344,73	\$ 1.038.888,84	\$ 944.419,05
10	PASIVO CORRIENTE	\$ 42.262.462,06	\$ 43.448.166,84	\$ 3.506.227,63	\$ 3.470.076,60
11	TOTAL PN Y PASIVO	\$ 168.156.029,21	\$ 168.353.214,25	\$ 14.217.012,84	\$ 12.747.077,50
12					
13	Tipo de cambio: 1 Euro =	\$1,2973			
14					